

```

> restart:
with(plots):
nmax := 20:
> Skalarprodukt := proc(f,g)
int(f*g,x=0..1)
end proc:
> for n from 1 to nmax do
f[n] := x^(n-1)
od:
> for n from 1 to nmax do
b[n] := f[n] - add(Skalarprodukt(f[n],b[k])*b[k],k=1..n-1);
b[n] := expand(b[n] / sqrt(Skalarprodukt(b[n],b[n])));
if n<=10 then print(b[n]) fi;
od:

```

$$\begin{aligned}
& 1 \\
& 2\sqrt{3}x - \sqrt{3} \\
& 6\sqrt{5}x^2 + \sqrt{5} - 6\sqrt{5}x \\
& 20\sqrt{7}x^3 - \sqrt{7} + 12\sqrt{7}x - 30\sqrt{7}x^2 \\
& 210x^4 - 420x^3 + 270x^2 - 60x + 3 \\
& 252\sqrt{11}x^5 - \sqrt{11} + 30\sqrt{11}x - 210\sqrt{11}x^2 + 560\sqrt{11}x^3 - 630\sqrt{11}x^4 \\
& 924\sqrt{13}x^6 + \sqrt{13} - 42\sqrt{13}x + 420\sqrt{13}x^2 - 1680\sqrt{13}x^3 + 3150\sqrt{13}x^4 \\
& \quad - 2772\sqrt{13}x^5 \\
& 3432\sqrt{15}x^7 - \sqrt{15} + 56\sqrt{15}x - 756\sqrt{15}x^2 + 4200\sqrt{15}x^3 - 11550\sqrt{15}x^4 \\
& \quad + 16632\sqrt{15}x^5 - 12012\sqrt{15}x^6 \\
& -72\sqrt{17}x + 1260\sqrt{17}x^2 - 9240\sqrt{17}x^3 + 34650\sqrt{17}x^4 - 72072\sqrt{17}x^5 \\
& \quad + 84084\sqrt{17}x^6 - 51480\sqrt{17}x^7 + 12870\sqrt{17}x^8 + \sqrt{17} \\
& 90\sqrt{19}x - 1980\sqrt{19}x^2 + 18480\sqrt{19}x^3 - 90090\sqrt{19}x^4 + 252252\sqrt{19}x^5 \\
& \quad - 420420\sqrt{19}x^6 + 411840\sqrt{19}x^7 - 218790\sqrt{19}x^8 + 48620\sqrt{19}x^9 - \sqrt{19}
\end{aligned}
\tag{1}$$

```

> A := Matrix([seq([seq(Skalarprodukt(b[j],b[k]),j=1..10]),k=1..10)])
;


```

(2)

$$A := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$


(2)

```
> display([seq(plot(b[n],x=0..1,y=-4..4,color=orange),n=1..nmax)],
insequence);
```



```
> # g := 1/(1+10*x);
# g := 1/(1+10*x^2);
# g := 1-abs(2*x-1);
g := (1+signum(2*x-1))/2;
plot(g,x=0..1,y=-0.5..1.5,color=blue);
```

$$g := \frac{1}{2} + \frac{\text{signum}\left(x - \frac{1}{2}\right)}{2}$$



```
> c := [seq(simplify(Skalarprodukt(g,b[n])),n=1..nmax)];
```

$$c := \left[\frac{1}{2}, \frac{\sqrt{3}}{4}, 0, -\frac{\sqrt{7}}{16}, 0, \frac{\sqrt{11}}{32}, 0, -\frac{5\sqrt{15}}{256}, 0, \frac{7\sqrt{19}}{512}, 0, -\frac{21\sqrt{23}}{2048}, 0, \frac{99\sqrt{3}}{4096}, 0, \right. \\ \left. -\frac{429\sqrt{31}}{65536}, 0, \frac{715\sqrt{35}}{131072}, 0, -\frac{2431\sqrt{39}}{524288} \right] \quad (3)$$

```
> for n from 1 to nmax do
h[n] := collect(expand(add(c[k]*b[k],k=1..n)),x)
od:
h[10];
```

$$\frac{3465}{256}x - \frac{45045}{128}x^2 + \frac{15015}{4}x^3 - \frac{5180175}{256}x^4 + \frac{7774767}{128}x^5 - \frac{13528515}{128}x^6 \\ + \frac{3391245}{32}x^7 - \frac{14549535}{256}x^8 + \frac{1616615}{128}x^9 - \frac{63}{512} \quad (4)$$

```
> display([seq(plot([g,h[n]],x=0..1,y=-0.5..1.5,color=[blue,red]),n=
1..nmax)], insequence);
```

